# $Id$
# Maintainer: AndyRTR <andyrtr@archlinux.org>
# Maintainer: Jan de Groot <jgc@archlinux.org>
# Contributor: Alexander Baldeck <alexander@archlinux.org>

pkgname=libxinerama
pkgver=1.1.3
pkgrel=2
pkgdesc="X11 Xinerama extension library"
arch=('i686' 'x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxext' 'xineramaproto')
makedepends=('xorg-util-macros')
source=("${url}/releases/individual/lib/libXinerama-${pkgver}.tar.bz2")
sha256sums=('7a45699f1773095a3f821e491cbd5e10c887c5a5fce5d8d3fced15c2ff7698e2')

build() {
  cd libXinerama-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd libXinerama-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
